



UNIVERSITY
OF COLOGNE

Data User Workshop: Natural Disasters and Entrepreneurship

Ole Lück (University of Cologne, Germany)

Institute of Geography (Working group: "Economic Geography and the Global South")

Today's agenda

- Natural Disasters and Entrepreneurship – an overview
- Coding goals and STATA output
- Overview of Datasets
- Climate shock indices
- First coding part: Exploration of self-employment dynamics
- Second coding part: Exploration of natural disasters
- Key Take-Aways

Natural Disasters and Entrepreneurship – an overview

- Micro firms build the backbone of local economies and often act as a poverty fighting strategy in developing and emerging countries (e.g. Mead and Liedholm 1998, Rijkers and Söderbom, 2013)
- Opportunity entrepreneurs can play a key role in establishing new growth paths, which was identified in earlier research using TVSEP data (Sohns and Revilla Diez 2018)
- Thailand and Vietnam are highly affected by Climate Change: Natural disasters are becoming more frequent and intense (e.g. IPCC 2022, Eckstein et al. 2020)
- Studies for other countries imply negative impacts on probability for households to be entrepreneurial active (Churchill et al. 2024) and reveal external macro-impacts and business-specific damages on micro firms (Hudecheck et al., 2020; Miklian and Hoelscher, 2022; Mawejje, 2024)

→ What is the impact of natural disasters on entrepreneurial households in Thailand and Vietnam?

Coding goals and STATA output

Goals:

- Understanding self-employment dynamics and exposure towards natural disasters across time and space
- Dealing with different spatial and panel datasets
- Creating basic and advanced graphs with STATA
- Setting the stage for future empirical analysis

STATA output:

- Bar charts showing self-employment dynamics in different provinces of Vietnam and Thailand
- Box plots showing drought exposure in Vietnam and Thailand (SPEI12) across Survey Waves
- Table showing storm exposure by province

Overview of Datasets

- Household and firm information: TVSEP Survey Data (2007, 2008, 2010, 2013, 2016, 2017, 2019, 2022)
- Weather Data:
 - Storm Track Data ([International Best Track Archive for Climate Stewardship \(IBTrACS\) Project](#))
 - High-resolution (5 km) gridded drought records: Precipitation: Climate Hazards Group InfraRed Precipitation with Station Data (CHIRPS, version 2); Evaporation: Global Land Evaporation Amsterdam Model (GLEAM, version 3.7a) → [SPEI data](#)
 - Extracted and merged with village coordinates in “R” → code available upon request

Climate shock indices

- Storms:

Based on “Exposure to weather shocks: A comparison between self-reported record and extreme weather data” Nguyen & Nguyen (2020) and Saffir-Simpson Hurricane Wind Scale: **Events having at least 33m/s and village are at least 250km close to event**

→ <https://www.nhc.noaa.gov/aboutsshws.php> (Saffir-Simpson Hurricane Wind Scale)

- Droughts: **SPEI12 index < -1.28 SD**

→ recommended by Agnew (2000)

SPEI	Probability	Category
>1.65	0.05	Extremely humid
>1.28	0.1	Severely humid
>0.84	0.2	Moderately humid
>-0.84 and <0.84	0.6	Normal
<-0.84	0.2	Moderately dry
<-1.28	0.1	Severely dry
<-1.65	0.05	Extremely dry

Source: Gebrechorkos et al. (2023)



UNIVERSITY
OF COLOGNE

FIRST PART: EXPLORATION OF SELF-EMPLOYMENT DYNAMICS

Constructing Sample Data I

- First step: **Clean data for each survey wave and country: create/rename variable v61001c for each wave (if necessary)**
- Use either general household data and merge with self-employment data or use only household data with self-employment variable being included ex-ante
- **v61001c** of the survey asks for engagement in Non-farm Self-Employment (must be generated for some older waves: e.g. in 2013)
- Account for different dataset types!

```
***** 2022 *****
*VN
use "$vietnam/2022/TVSEP2022SurveyV1.dta", clear
keep QID v61001c v10003 v10004
destring QID, replace
format QID %15.0f
gen year = 2022
save "$vietnam/2022/TVSEP2022SurveyV1_selfemplonly.dta", replace

*TH
use "$thailand/2022/TVSEP2022SurveyV1.dta", clear
keep QID v61001c v10003 v10004
gen year = 2022
save "$thailand/2022/TVSEP2022SurveyV1_selfemplonly.dta", replace
```

```
***** 2013 *****
*VN
use "$vietnam/2013/hhclean.dta", clear
tostring QID, replace format(%15.0f)
save "$vietnam/2013/hhclean_stringQID.dta", replace

use "$vietnam/2013/selfemplclean.dta", clear
keep QID
duplicates drop QID, force
gen v61001c = 1
merge 1:m QID using "$vietnam/2013/hhclean_stringQID.dta"
replace v61001c = 2 if _merge==2
gen year = 2013
keep QID v61001c year
destring QID, replace
save "$vietnam/2013/hhclean_selfemplonly.dta", replace
```


Constructing Sample Data II

- Append household/self-employment data of each wave for Thailand and Vietnam
- Reshape function to transform wide format to long format -> better suited data format for panel analysis
- **v61001c2007, v61001c2008, ...** → variables showing if HH mentioned to have at least one micro firm during the reference period of Survey Wave

```
*****
* merging all waves VN
*****
use "$vietnam/2022/TVSEP2022SurveyV1_selfemplonly.dta", clear
append using "$vietnam/2017/hhclean_selfemplonly.dta"
append using "$vietnam/2016/hhclean_selfemplonly.dta"
append using "$vietnam/2013/hhclean_selfemplonly.dta"
append using "$vietnam/2010/hhclean_selfemplonly.dta"
append using "$vietnam/2008/hhclean_selfemplonly.dta"
append using "$vietnam/2007/hhclean_selfemplonly.dta"
sort QID year
order QID year
format QID %15.0f

reshape wide v61001c v10003 v10004 subdistr vill _x10003 _x10004, i(QID) j(year)

* missing year means that HH information IN GENERAL is not available

keep QID v61001c*

gen country = "VN"
save "$vietnam/Aggregates/selfemplonly_07to22_QID_VN.dta", replace
```

```
*****
* merging all waves VN + TH
*****

use "$vietnam/Aggregates/selfemplonly_07to22_QID_VN.dta", clear
append using "$thailand/Aggregates/selfemplonly_07to22_QID_TH.dta"

order QID country v61001c2007 v61001c2008 v61001c2010 v61001c2013 v61001c2016 v61001c2017 v61001c2019 v61001c2022
save "$masterfolder/Data_Aggregates/selfemplonly_07to22_QID_VN_TH.dta", replace
```

Constructing Sample Data III

```
* load weights-data in order to integrate village and year specific weights and to get province information
use "$masterfolder/Weights/Weights_TH_VN.dta", clear

destring QID, replace
format QID %15.0f

merge 1:1 QID using "$masterfolder/Data_Aggregates/selfemplonly_07to22_QID_VN_TH.dta"

drop if _merge == 1
drop _merge

order QID v61001c2007 v61001c2008 v61001c2010 v61001c2013 v61001c2016 v61001c2017 v61001c2022


reshape long v61001c, i(QID) j(year)
drop if v61001c == .
drop if v61001c == 3
```

To get consistent data on location of HH and weights to account for scaling results
→ merge dataset with “*Weights_TH_VN.dta*” (BUT: weights are not used today)

Creation of Bar Charts

Task: Show how often HH are engaged into Non-farm Self-employment across all Survey Waves in which they occur in TVSEP – Proportion of HH showing certain engagement behavior

- a) Create bar chart on country-level (for Thailand and Vietnam)
 - b) Create bar chart on provincial-level (for all six provinces in both countries)
- Make use of the *histogram* command
 - Make use of “binned” groups of users




```
gen has_firm = .
replace has_firm = 1 if v61001c == 1
replace has_firm = 0 if v61001c == 2

bysort QID (year): gen year_count = _N if _n == 1



egen sum_has_firm = sum(has_firm), by(QID)
gen relative_years_used = (sum_has_firm/year_count) *100

* binning relative years for better overview
gen relative_years_used_bins = .
replace relative_years_used_bins = 0 if relative_years_used == 0
replace relative_years_used_bins = 1 if relative_years_used <= 50 & relative_years_used > 0
replace relative_years_used_bins = 2 if relative_years_used < 100 & relative_years_used > 50
replace relative_years_used_bins = 3 if relative_years_used == 100
```

Vietnam



Thailand



Solution

```
*Vietnam*

* country
  histogram relative_years_used_bins if country == "VN", horizontal discrete fraction ///
  xlabel(0(0.1)0.5) ///
  ylabel(0 "Never self-employed" 1 "<= 50% of Survey Waves self-employed" 2 "> 50% of Survey Waves self-employed" ///
  3 "Always self-employed", angle(0)) ///
  color(gray) ///
  scheme(s2color) ///
  lcolor(black) ///
  xtitle("Fraction of Households", size(small)) ///
  ytitle("") ///
  barwidth(1)

graph export "$masterfolder/Output/Self_Employment_VN_country.png", as(png) replace

* provinces
  histogram relative_years_used_bins if country == "VN", horizontal discrete fraction by(_x10001, row(1) title("")) ///
  xlabel(0(0.1)0.5) ///
  ylabel(0 "Never self-employed" 1 "<= 50% of time self-employed" 2 "> 50% of time self-employed" ///
  3 "Always self-employed", angle(0)) ///
  color(gray) ///
  scheme(s2color) ///
  lcolor(black) ///
  xtitle("Fraction of Households", size(small)) ///
  ytitle("") ///
  barwidth(1)

graph export "$masterfolder/Output/Self_Employment_VN_provinces.png", as(png) replace
```



UNIVERSITY
OF COLOGNE

SECOND PART: EXPLORATION OF NATURAL DISASTERS

Creation of box plots - Droughts

Task: Visualize SPEI12 index for the year (reference period) of each Survey Wave

- a) cumulative for all households
 - b) only for Vietnamese households
 - c) only for Thai households
-
- Make use of the *graph hbox* command


Solution

```
* DROUGHTS *
*SPEI 12 by country

*pooled
graph hbox spei12, by(year,cols(1) title("Pooled", size(mediumsmall)) note("")) yline(-1.28, lwidth(thin) lpattern(solid) lcolor(orange)) ///
  yline(-1.65, lwidth(thin) lpattern(solid) lcolor(red)) name(pooled, replace) scheme(s2mono)
*TH
graph hbox spei12 if T == 1, by(year,cols(1) title("Thailand", size(mediumsmall)) note("")) yline(-1.28, lwidth(thin) lpattern(solid) lcolor(orange)) ///
  yline(-1.65, lwidth(thin) lpattern(solid) lcolor(red)) name(T, replace) scheme(s2mono)
*VN
graph hbox spei12 if T == 0, by(year,cols(1) title("Vietnam",size(mediumsmall)) note("")) yline(-1.28, lwidth(thin) lpattern(solid) lcolor(orange)) ///
  yline(-1.65, lwidth(thin) lpattern(solid) lcolor(red)) yscale(range(-3 2)) name(V, replace) scheme(s2mono)

*combine graphs
graph combine pooled T V, ycommon nocopies rows(1)

graph export "$masterfolder/Output/SPEI12_TH_VN.png", as(png) replace
```



Creation of Table - Storms

Task: Table showing storm-affected households by provinces and year

- Make use of *tab* command

Solution

Province ID	year							Total
	2007	2010	2013	2016	2017	2019	2022	
Dak Lak	6	0	0	0	76	76	0	158
Ha Tinh	72	66	72	72	72	0	0	354
Nakhon Phanom	40	0	40	0	40	0	0	120
Thua Thien Hue	72	24	72	0	72	0	72	312
Ubon Ratchathani	16	0	20	0	18	0	67	121
Total	206	90	204	72	278	76	139	1,065

No storm-affected HH in Buriram

Key Take-Aways

Examining household engagement in micro-firms over time provides insights into the volatility of non-farm self-employment and the potential vulnerability of these businesses:

- Engagement into Non-farm Self-Employment is highly volatile
- Micro firms usually do not sustain long, suggesting lack of resilience (?) → applies for each of the six provinces in both countries
- Severe weather events are present during all survey years implying higher a risk of (economic) disruptions
- Shock exposure differs in type, intensity and location
- In future: use specific firm variables (Section 6 in TVSEP survey) to proceed with analysis on heterogeneous, group-specific effects

References

Agnew, C. T.: Using the SPI to Identify Drought, Drought Network News (1994–2001), <https://digitalcommons.unl.edu/droughtnetnews/1>, 2000.

Awaworyi Churchill, S., Munyanyi, M. E., Trinh, T.-A., & Wiklund, J. (2024). From disastrous heat waves to extreme rains: Effects of weather shocks on entrepreneurship. *Journal of Business Venturing Insights*, 21, Article e00469. <https://doi.org/10.1016/j.jbvi.2024.e00469>

Eckstein, D., Künzel, V., Schäfer, L., & Winges, M. (2020). Global Climate Risk Index 2020: Who Suffers Most from Extreme Weather Events? Weather-Related Loss Events in 2018 and 1999–2018. Germanwatch e.V. (Report). Retrieved from <https://www.germanwatch.org/en/17307>

Gebrechorkos, S. H., Peng, J., Dyer, E., Miralles, D. G., Vicente-Serrano, S. M., Funk, C., Beck, H. E., Asfaw, D. T., Singer, M. B., and Dadson, S. J.: Global high-resolution drought indices for 1981–2022, *Earth Syst. Sci. Data*, 15, 5449–5466, <https://doi.org/10.5194/essd-15-5449-2023>, 2023.

Hudecheck, M., Sirén, C., Grichnik, D., & Wincent, J. (2020, March 9). How companies can respond to the coronavirus. MIT Sloan Management Review.

IPCC. (2022). *Climate Change 2022: Impacts, Adaptation and Vulnerability*. Contribution of Working Group II to the Sixth Assessment Report of the Intergovernmental Panel on Climate Change (H.-O. Pörtner, D. C. Roberts, M. Tignor, et al., Eds.). Cambridge University Press. <https://doi.org/10.1017/9781009325844>

References

- Hudecheck, M., Sirén, C., Grichnik, D., & Wincent, J. (2020, March 9). How companies can respond to the coronavirus. MIT Sloan Management Review. Retrieved from <https://sloanreview.mit.edu/article/how-companies-can-respond-to-the-coronavirus/>
- IPCC. (2022). Climate Change 2022: Impacts, Adaptation and Vulnerability. Contribution of Working Group II to the Sixth Assessment Report of the Intergovernmental Panel on Climate Change (H.-O. Pörtner, D. C. Roberts, M. Tignor, et al., Eds.). Cambridge University Press. <https://doi.org/10.1017/9781009325844>
- Mawejje, J. (2024). How does the weather and climate change affect firm performance in low-income countries? Evidence from Uganda. *Sustainable Futures*, 7, 100167. <https://doi.org/10.1016/j.sftr.2024.100167>
- Miklian, J., & Hoelscher, K. (2022). SMEs and exogenous shocks: A conceptual literature review and forward research agenda. *International Small Business Journal*, 40(2), 178–204. <https://doi.org/10.1177/02662426211050796>
- Mead, D. C., & Liedholm, C. (1998). The dynamics of micro and small enterprises in developing countries. *World Development*, 26(1), 61–74. [https://doi.org/10.1016/S0305-750X\(97\)10010-9](https://doi.org/10.1016/S0305-750X(97)10010-9)
- Rijkers, B., & Söderbom, M. (2013). The effects of risk and shocks on non-farm enterprise development in rural Ethiopia. *World Development*, 45, 119–136. <https://doi.org/10.1016/j.worlddev.2012.10.013>
- Sohns, F., & Revilla Diez, J. (2018). Explaining micro entrepreneurship in rural Vietnam—a multilevel analysis. *Small Business Economics*, 50(1), 219–237. <https://doi.org/10.1007/s11187-017-9886-2>